(DEFUN datos_txcurva (/ mens lconj grad prop tent)

  (IF altx0 () (SETQ altx0 1))

  (SETQ mens (STRCAT "Altura del texto <" (RTOS altx0 2 2) ">: "))

  (INITGET 6)

  (IF (SETQ altx (GETREAL mens)) () (SETQ altx altx0))(TERPRI)

  (SETQ altx0 altx)

  (IF sep0 () (SETQ sep0 0))

  (SETQ mens (STRCAT "Separación entre texto y entidad (+ = encima;- = debajo)"

                     " <" (RTOS sep0 2 2) ">: "))

  (IF (SETQ sep (GETREAL mens)) ( ) (SETQ sep sep0))(TERPRI)

  (SETQ sep0 sep)

  (IF (NOT (TBLSEARCH "block" "$txcurva"))

    (PROGN (COMMAND "_point" "0,0")

    (COMMAND "_block" "$txcurva" "0,0" (entlast) ""))

  )

  (SETQ prop (CDR (ASSOC 41 (TBLSEARCH "style" (GETVAR "textstyle")))))

  (SETQ esp (* altx prop 1.2))

  (SETQ mens "Designar entidad para alinear texto: ") 

  (WHILE (NOT (SETQ grad (ENTSEL mens)))(SETQ grad (ENTSEL mens)))

  (SETQ tent (CDR (ASSOC 0 (ENTGET (CAR grad)))))

  (WHILE (NOT (OR (= tent "ARC")(= tent "CIRCLE")(= tent "ELLIPSE")

                  (= tent "LINE")(= tent "POLYLINE")(= tent "LWPOLYLINE")



      (= tent "SPLINE")))

    (PROMPT "\nEntidad no válida para situar texto\n")

    (WHILE (NOT (SETQ grad (ENTSEL mens)))(SETQ grad (ENTSEL mens)))

    (SETQ tent (CDR (ASSOC 0 (ENTGET (CAR grad)))))

  )

  (COMMAND "_measure" grad "_b" "$txcurva" "_y" esp)(TERPRI)

  (SETQ conj (SSGET "_p"))

  (SETQ lconj (SSLENGTH conj) lontx (+ lconj 1))

  (WHILE (> lontx lconj)

    (PROMPT (STRCAT "Número máximo de caracteres: " (ITOA lconj) "\n"))

    (SETQ cadtx (GETSTRING T "Introducir Texto: "))(TERPRI)

    (SETQ lontx (STRLEN cadtx))

  )

)

(DEFUN txcurva (/ ntx desig tx nent lent pin rot )

  (SETQ ntx 0 desig (SSADD))

  (REPEAT lontx

    (SETQ tx (SUBSTR cadtx (+ ntx 1) 1))

    (SETQ nent (SSNAME conj ntx))

    (SETQ lent (ENTGET nent))

    (SETQ pin (CDR (ASSOC 10 lent)))

    (SETQ rot (CDR (ASSOC 50 lent)))

    (IF (OR (= tx "l")(= tx "i")(= tx "I")(= tx "j")(= tx "1"))

      (SETQ pin (POLAR pin rot (/ altx 6))))

    (IF (= tx "m")

      (SETQ pin (POLAR pin rot (- (/ altx 6)))))

    (SETQ pin (POLAR pin (+ rot (/ PI 2)) sep))            

    (SETQ rot (/ (* rot 180) PI))

    (COMMAND "_text" pin altx rot tx)

    (SETQ desig (SSADD (SSNAME (SSGET "_l") 0) desig))

      (SETQ ntx (+ ntx 1))

  )

  (COMMAND "_erase" conj "")

  (COMMAND "_select" desig "")

  (COMMAND "_redraw")

)

(DEFUN c:txcurva (/ altx sep esp conj lontx cadtx refnt0 pdmod0 error0) 

  (SETVAR "cmdecho" 0)

  (SETQ error0 *error* *error* err_txcurva)

  (SETQ pdmod0 (GETVAR "pdmode")) (SETVAR "pdmode" 0)

  (SETQ refnt0 (GETVAR "osmode"))(SETVAR "osmode" 0)   

  (COMMAND "_undo" "_begin")

  (datos_txcurva)

  (txcurva)

  (COMMAND "_purge" "_block" "$txcurva" "_n")

  (COMMAND "_undo" "_end")

  (SETVAR "osmode" refnt0)(SETVAR "pdmode" pdmod0)

  (SETVAR "cmdecho" 1)(PRIN1)

)

(DEFUN err_txcurva (mens)

  (SETQ *error* error0)

  (IF (= mens "quitar / salir abandonar")

    (PRIN1)

    (PRINC (STRCAT "\nError: " mens " "))

  )

  (IF conj (COMMAND "_erase" conj ""))

  (COMMAND "_purge" "_block" "$txcurva" "_n")

  (COMMAND "_undo" "_end")

  (SETVAR "cmdecho" 1)

  (SETVAR "osmode" refnt0)(SETVAR "pdmode" pdmod0)(PRIN1)

)

(PROMPT "Nuevo comando TXCURVA definido.")(PRIN1)

